Circuit Analysis with your TRS-80 Computer

Here are five programs to simplify electronic design work.

BY ROY BABYLON

AC Formulas



THE first program in this group deals with six areas of general interest in ac circuit design. The first three have to do with the relationship among peak voltage, instantaneous voltage, and the angle of the sine wave. The next two involve the relationship between frequency and wavelength. Anyone working with tuned circuits or antennas will find these sections useful. A unique feature of the program is that it gives answers in both feet and meters. This part is also helpful in converting scope readings. Normally measured in time (period), these readings are converted into the more familiar and useful frequency equivalents. The final part of the program handles conversion between the various relationships of rms, peak, peak-to-peak, and average amplitude values.

The program was converted from a Level I to a Level II format (which explains its layout), and it is directly compatible with either one. The main difference between the two formats is that Level II has subroutines such as square root, angle functions, etc., built in. This is why all the subroutines were placed at the end of the program. For a Level I machine, enter the program as written. It must be modified accordingly for a Level II machine. ◇

```
3 CLS
20  PRINTTAB(15);"VARIOUS A.C. FORMULAS":PRINT
35 PRINT "ENTER NUMBER NEXT TO UNKNOWN:":PRINT
40 PRINT "PEAK VOLTS=1",,"PERIOD-FREQUENCY=4"
41 PRINT "INSTANTANEOUS VOLTAGE=2","WAVELENGTH (METERS OR FEET)=5"
43 PRINT "ANGLE POINT ON WAVE=3","RMS-PEAK-AVERAGE CONVERSIONS=6":PRINT
50  INPUT"REQUESTED FORMULA IS ";U
60   IF U=1 GOTO90
65   IF U=2 GOTO140
70   IF U=3 GOTO410
75   IF U=4 GOTO460
80   IF U=5 GOTO560
85    IF U=6 GOTO190
90  INPUT"ENTER INSTANTANEOUS VOLTAGE";I
100 INPUT"ENTER ANGLE IN DEGREES";X
110  GOSUB 30376
120 PRINT"PEAK VOLTAGE EQUALS ";I/Y;"VOLTS"
130  END
140 INPUT"ENTER PEAK VOLTAGE IN VOLTS";E
150 INPUT"ENTER ANGLE IN DEGREES";X
160  GOSUB30376
170 PRINT"THE INSTANTANEOUS VOLTAGE IS";E*Y;"VOLTS"
180 END
190 PRINT"ENTER THE NUMBER NEXT TO THE KNOWN VOLTAGE"
200 PRINT"RMS (EFFECTIVE) VOLTAGE 1"
210 PRINT"PEAK VOLTAGE            2"
220 PRINT"AVERAGE VOLTAGE         3"
230 INPUT"THE KNOWN VOLTAGE IS";U
240 IF U=1 GOTO270
250 IF U=2 GOTO310
260 IF U=3 GOTO360
270 INPUT"ENTER THE RMS VOLTAGE";X
280 PRINT"THE PEAK TO PEAK VOLTAGE IS";2.828*X;"VOLTS"
290 PRINT"THE PEAK VOLTAGE IS";1.414*X;"VOLTS"
300 PRINT"THE AVERAGE VOLTAGE IS";.9*X;"VOLTS"
305 END
310 INPUT"ENTER PEAK VOLTS";P
320 PRINT"THE PEAK TO PEAK VOLTAGE IS";2*P;"VOLTS"
330 PRINT"THE RMS VOLTAGE IS";.707*P;"VOLTS"
340 PRINT"THE AVERAGE VALUE IS";.636*P;"VOLTS"
345 END
360 INPUT"ENTER AVERAGE VOLTAGE";A
370 PRINT"THE PEAK VOLTAGE IS";1.57*A;"VOLTS"
380 PRINT"THE PEAK TO PEAK VOLTAGE IS";3.14*A;"VOLTS"
390 PRINT"THE RMS VOLTAGE IS";1.11*A;"VOLTS"
400 END
410 INPUT"ENTER PEAK VOLTAGE";P
420 INPUT"ENTER INSTANTANEOUS VOLTAGE";I
430  S=I/P
435  GOSUB 30550
440 PRINT"THE ANGLE ON THE SINE WAVE IS";Y;"DEGREES"
442 END
444 INPUT"ENTER PEAK VOLTAGE";P
445 INPUT"ENTER PHASE ANGLE IN DEGREES";X
446  GOSUB 30376
448 PRINT"THE INSTANTANEOUS VOLTAGE IS";P*Y;"VOLTS"
450 END
460 PRINT" PERIOD=1 , FREQUENCY=2 "
470 INPUT"FORMULA DESIRED IS";X
480 IF X=1 GOTO 500
490 IF X=2 GOTO 530
500 INPUT"ENTER FREQUENCY IN HERTZ";F
510 PRINT"THE PERIOD OF";F;"HERTZ IS";1/F;"SECONDS"
520 END
530  INPUT"ENTER PERIOD IN SECONDS";T
540 PRINT"THE FREQUENCY IS";1/T;"HERTZ"
550 END
560 PRINT" WAVELENGTH=1 , FREQUENCY=2 "
570 INPUT"DESIRED FUNCTION ?";X
580 IF X=1 GOTO 600
590 IF X=2 GOTO 650
600 INPUT"ENTER F IN HERTZ";F
610  V=3E8/F
620   X=984E8/F
630 PRINT"THE WAVELENGTH(LAMBDA)IS EQUAL TO";V;"METERS OR";X;"FEET"
640 END
650 INPUT"ENTER WAVELENGTH";W
660 INPUT" METERS=1 , FEET=2 ";C
670 IF C=1 GOTO 690
680 IF C=2 GOTO 700
690 PRINT"FREQUENCY=";3E8/W;"HERTZ"
695 END
700 PRINT"THE FREQUENCY OF";W;"FEET IS";984E8/W;"HERTZ"
710 END
30376  Z=ABS(X)/X:X=Z*X
30380  IF X>360 THEN X=X/360 : X=(X-INT(X))*360
30390  IFX>90THENX=X/90:Y=INT(X):X=(X-Y)+90-ONYGOTO30410,30420,30430
30400  X=X/57.29578 :IF ABS(X)<2.48616E-4 Y=0:RETURN
30405  GOTO30440
30410  X=90-X : GOTO 30400
30420  X=-X : GOTO 30400
30430  X=X-90 : GOTO 30400
30440  Y=X-X*X*X/6+X*X*X*X*X*X/120-X*X*X*X*X*X*X/5040
30455  RETURN
30500  REM *ARCCOS* INPUT S, OUTPUT Y,W
30510  REM Y IS IN DEGREES, W IS IN RADIANS
30520  GOSUB 30550 : Y=90-Y : W= 1.570796-W : RETURN
30550  X=S : IF ABS(S)<=.707107 THEN 30610
30560  X=1-S*S : IF X<0 THEN PRINT S; "IS OUT OF RANGE" : STOP
30570  W=X/2 : Z=0
30580   Y=(X/W-W)/2 : IF (Y=0)+(Y=Z) THEN X=W : GOTO 30610
30600  W=W+Y : Z=Y : GOTO30580
30610  Y=X+X*X*X/6+X*X*X*X*X*X*.075+X*X*X*X*X*X*X*X*4.464286E-2
30620  W=Y+X*X*X*X*X*X*X*X*X*3.038194E-2
30625   IF ABS(S)>.707107 THEN W=1.570796E-2
30630  Y=W*57.29578 : RETURN
```

Impedance Matching

THIS short program for the TRS-80 Level II was designed to furnish the proper values of resistors for an L-type impedance-matching network and also draws the circuit diagram and figures out the system loss in decibels.

The program is self-prompting and uses no subroutines. Lines 20 and 30 ask the designer for the two impedances to be matched. Lines 40 through 70 are the "working" lines and operate on the data supplied in lines 20 and 30. The working lines are self-explanatory. Lines 90 through 130 print component identification and values for the circuit drawn by lines 140 and 150. Line 140 draws all the horizontal lines, and line 150 takes care of the vertical. Line 160 is a simple clock, which keeps the drawing on the screen for a predetermined amount of time. With the value used, the drawing remains for about 35 seconds. (To increase this time, increase the value of "N.") After the clock in line 160 times out, the computer goes to line 170 which returns it to line 10 to await the next problem. ◇

```
10 CLS
15 PRINTTAB(15)"L TYPE IMPEDANCE MATCHING NETWORK"
20 INPUT"ENTER INPUT IMPEDANCE";Z1
30 INPUT"ENTER OUTPUT IMPEDANCE";Z2
40 R1=Z1*SQR(1-(Z2/Z1))
50 R2=Z2/SQR(1-(Z2/Z1))
60 M=SQR(Z1/Z2)+SQR((Z1/Z2)-1)
70 L=20*(LOG(M)/LOG(10))
80 CLS
90 PRINT@520,"Z IN >>>"
100 PRINT@205,"R1=";R1;"OHMS"
110 PRINT@560,"Z OUT <<<"
120 PRINT@920,"LOSS IN DBS IS";L
130 PRINT@595,"R2=";R2;"OHMS"
140 Y=15:FORX=10TO30:SET(X,Y):NEXTX:Y=12:FORX=30TO58:SET(X,Y):NEXTX:Y=18:FORX=30TO58:SET(X,Y):NEXTX:Y=15:FORX=58TO100:SET(X,Y):NEXTX
:Y=21:FORX=74TO87:SET(X,Y):NEXTX:Y=32:FORX=74TO87:SET(X,Y):NEXTX:Y=36:FORX=10TO100:SET(X,Y):NEXTX:
150 X=30:FORY=12TO18:SET(X,Y):NEXTY:X=58:FORY=12TO18:SET(X,Y):NEXTY:X=80:FORY=15TO21:SET(X,Y):NEXTY:X=74:FORY=21TO32:SET(X,Y):NEXTY:
X=87:FORY=21TO32:SET(X,Y):NEXTY:X=80:FORY=32TO36:SET(X,Y):NEXTY:
160 FORN=1TO10000:NEXTN
170 GOTO 10
```

Capacitance Networks

3.

HERE is a Level II program covering eight different areas in the design of capacitance networks. The first three areas are of primary interest to students taking a course in electronics and will probably be of little use to experimenters. The other five areas, however, will be useful for both groups.

The program is self-explanatory and self-prompting when run. One unique feature of the program (section 8) allows you to change the total capacitance of a network by a specific amount. This section of the program not only tells you what value to add, but also tells whether to put the added capacitance in series or parallel. ◇

```
10 CLS
20 PRINTTAB(15)"VARIOUS CAPACITIVE FORMULAE"
30 PRINT"CHARGE ON A CAPACITOR                    #1"
40 PRINT"VALUE OF A CAPACITOR (PHYSICAL FACTORS)  #2"
50 PRINT"ENERGY IN AN ELECTROSTATIC FIELD         #3"
60 PRINT"RESISTOR & CAPACITOR TIME CONSTANTS      #4"
70 PRINT"SERIES & PARALLEL REACTANCES             #5"
80 PRINT"CAPACITIVE REACTANCE                     #6"
90 PRINT"IMPEDANCE                                #7"
100 PRINT"SERIES/PARALLEL CAPACITORS              #8"
120 INPUT U
130 ON U GOTO 135,290,335,375,515,565,685,645
135 CLS
140 INPUT"ENTER UNKNOWN Q=1 , C=2 , E=3";X
150 ON X GOTO 160,210,250
160 INPUT"ENTER CAPACITOR VALUE IN U-FARADS";C
170 INPUT"ENTER VOLTAGE";V
180 PRINT"THE CHARGE IS";(C*V)/(1E6);"COULOMBS OF ELECTRONS "
190 END
210 INPUT"ENTER VOLTAGE";V
220 INPUT"ENTER CHARGE IN COULOMBS";Q
230 PRINT"THE CAPACITANCE IS";(Q/V)*(1E6),"MICROFARADS"
240 END
250 INPUT"ENTER CAPACITANCE IN U-FARADS";C
260 INPUT"ENTER CHARGE (Q) IN COULOMBS";Q
270 PRINT"THE VOLTAGE IS";(Q/C)*1E6,"VOLTS"
280 END
290 INPUT"ENTER AREA OF PLATES IN SQ. METERS";A
300 INPUT"ENTER DISTANCE BETWEEN PLATES IN METERS";D
310 INPUT"ENTER DIELECTRIC CONSTANT (FROM TABLE)";K
320 PRINT"THE CAPACITOR VALUE IS";K*(A/D)*8.85;"U-U FARADS"
330 END
335 CLS
340 INPUT"ENTER THE VOLTAGE VALUE";V
350 INPUT"ENTER CAPACITOR VALUE IN U-FARADS";C
360 PRINT"THE ENERGY IN THE ELECTROSTATIC FIELD IS";1/(2*C)*(V(2)*(1E6),"JOULES"
370 END
375 CLS
380 INPUT"ENTER UNKNOWN , TIME=1 , RESISTOR=2 , CAPACITOR=3";X
390 ON X GOTO 400,440,480
400 INPUT"ENTER RESISTOR VALUE IN OHMS";R
410 INPUT"ENTER CAPACITOR VALUE IN U-FARADS";C
420 PRINT"THE TIME CONSTANT IS";R*C*1E-6,"SECONDS"
430 END
435 CLS
440 INPUT"ENTER TIME IN SECONDS";T
450 INPUT"ENTER CAPACITOR IN U-FARADS";C
460 PRINT"THE RESISTOR VALUE IS";T/C*1E6,"OHMS"
470 END
475 CLS
480 INPUT"ENTER TIME IN SECONDS";T
490 INPUT"ENTER RESISTANCE IN OHMS";R
500 PRINT"THE CAPACITOR VALUE IS";T/R*1E6,"U-FARADS"
510 END
515 CLS
520 INPUT"ENTER CAPACITIVE REACTANCE (XC1)";A
530 INPUT"ENTER (XC2)";B
540 PRINT"THE SERIES REACTANCE IS";A+B;"OHMS CAPACITIVE"
550 PRINT"THE PARALLEL REACTANCE IS";(A*B)/(A+B);"OHMS"
560 END
565 CLS
570 INPUT"ENTER CAPACITOR VALUE IN U-FARADS";C
580 INPUT"ENTER FREQUENCY IN HERTZ";F
590 PRINT"THE REACTANCE IS";(1/(6.28*F*C*1E-6));"OHMS OF XC"
600 END
605 CLS
610 INPUT"ENTER REACTANCE (XC) IN OHMS";B
620 INPUT"ENTER RESISTANCE IN OHMS";R
630 PRINT"IMPEDANCE (Z) EQUALS";SQR((BC2)+(RC2));"OHMS"
640 END
645 CLS
650 INPUT"ENTER 1 FOR TOTAL VALUE , 2 FOR CHANGING TOTAL VALUE";Q
660 ON Q GOTO 670,840
665 CLS
670 INPUT"ENTER NUMBER OF CAPACITORS USED,   2 , 3 , OR 4";X
680 ON X GOTO,685,725,775
685 CLS
690 INPUT"ENTER VALUE OF C1";A
700 INPUT"ENTER VALUE OF C2";B
710 PRINT"THE SERIES VALUE IS";(A*B)/(A+B);"AND THE PARALLEL VALUE IS";A+B;"U-FARADS"
720 END
725 CLS
730 INPUT"ENTER VALUE OF C1";A
740 INPUT"ENTER VALUE OF C2";B
750 INPUT"ENTER VALUE OF C3";C
760 PRINT"THE SERIES VALUE IS";1/(1/A+1/B+1/C);"AND THE PARALLEL VALUE IS";A+B+C;"U-FARADS"
770 END
775 CLS
780 INPUT"ENTER VALUE OF C1";A
790 INPUT"ENTER VALUE OF C2";B
800 INPUT"ENTER VALUE OF C3";C
810 INPUT"ENTER VALUE OF C4";D
820 PRINT"THE SERIES VALUE IS";1/(1/A+1/B+1/C+1/D);"AND THE PARALLEL VALUE IS";A+B+C+D;"U-FARADS"
830 END
835 CLS
840 INPUT"ENTER PRESENT TOTAL VALUE OF CAPACITANCE";T
850 INPUT"ENTER DESIRED VALUE";D
860 IF T>D GOTO 870 ELSE 880
870 PRINT"TO OBTAIN";D;"AS A FINAL VALUE , ADD";((T*D)/(T-D));"IN SERIES WITH  YOUR PRESENT CIRCUIT"
875 END
880 PRINT"TO OBTAIN";D;"AS A FINAL VALUE  ADD";D-T;"IN PARALLEL WITH YOUR PRESENT CIRCUIT."
890 END
```


Common-Emitter Amp Design

4.

IF YOU are designing a common-emitter amplifier such as that shown in the diagram and have only minimum information about the transistor, this is a useful program to have. The only data needed is the maximum collector current, maximum collector voltage, beta, and type of transistor (silicon or germanium). The layout of the program is straightforward and it is self-prompting when run.



Lines 30 through 70 ask for information needed for the calculations in lines 80 through 140. Lines 150 through 190 print out the results of the calculations and label what they are. These values are the "ideal" and it is up to the designer to pick the closest standard values available.

The remainder of the program draws a schematic of the circuit and labels all parts values. Line 200 begins by asking if a schematic is necessary. If not, the program jumps to line 220 which ends it. If a schematic is desired, the program jumps to line 230, which begins the graphics by clearing the screen and then printing the parts values and labels. Lines 250 through 310 draw the rest of the schematic.

Line 320 is a simple clock. With the value of "N" used, the graphics remain on the screen for about 20 seconds. To increase this time, make "N" a larger number. After the clock times out, the screen is cleared and the program is sent back to line 20 to await the next problem. ◇

```
20 PRINT"COMMON EMITTER AMPLIFIER"
30 INPUT"ENTER MAX. COLLECTOR CURRENT";A
40 INPUT"ENTER BETA";B
50 INPUT"ENTER VCC VOLTAGE";C
60 INPUT"ENTER TYPE: SILICON=1 , GERMANIUM=2";D
70 INPUT"ENTER LOWEST FREQUENCY TO BE AMPLIFIED";K
80 E=.5*A'IC
90 F=.45*C'VC
100 G=.1*C'VE
110 I=E/B'IB
120 IF D=1 H=G+.6 ELSE H=G+.3'VB
130 J=(C-F)/I 'R1
140 L=.1*(G/E):M=(1/(6.28*K*L))*1E6
150 PRINT"R1=";J;"OHMS"
160 PRINT"R2=";(G/C)*J;"OHMS"
170 PRINT"R3=";(C-F)/E;"OHMS
180 PRINT"R4=";G/E;"OHMS"
190 PRINT"CI=";M;"MICRO FARADS"
200 INPUT"IS A SCHEMATIC DESIRED ? YES=1 , NO=2";R
210 IF R=1 GOTO230ELSE 220
220 END
230 CLS
240 PRINT@204,"R1":PRINT@218,"R3":PRINT@19,"VCC":PRINT@349,"C":PRINT@469,"B":PRINT@540,"E":PRINT@716,"R2":PRINT@730,"R4":PRINT@743,"
C1 =":PRINT@914,"GROUND":PRINT@192,;J;:PRINT@210,;(C-F)/E;:PRINT@705,;.25*J;:PRINT@723,;G/E;:PRINT@747,;M;" MICRO FARADS"
250 X=40:FORY=17TO26:SET(X,Y):NEXTY:
260 X=40:Y=20:SET(X,Y):X=40:Y=20:FORN=1TO10:SET(X,Y):X=X+1.5:Y=Y-.5:NEXTN:X=53:FORY=14TO16:SET(X,Y):NEXTY:X=53:FORY=28TO30:SET(X,Y):
NEXTY:X=40:Y=24:SET(X,Y):X=40:Y=24:FORN=1TO10:SET(X,Y):X=X+1.5:Y=Y+.5:NEXTN:
270 Y=4:FORX=26TO53:SET(X,Y):NEXTX:Y=6:FORX=22TO30:SET(X,Y):NEXTX:Y=6:FORX=49TO57:SET(X,Y):NEXTX:Y=14:FORX=22TO30:SET(X,Y):NEXTX:Y=1
4:FORX=49TO57:SET(X,Y):NEXTX:Y=22:FORX=26TO40:SET(X,Y):NEXTX:Y=30:FORX=22TO30:SET(X,Y):NEXTX:
280 Y=30:FORX=49TO57:SET(X,Y):NEXTX:Y=28:FORX=53TO69:SET(X,Y):NEXTX:Y=33:FORX=64TO74:SET(X,Y):NEXTX:Y=35:FORX=64TO74:SET(X,Y):NEXTX:
Y=30:FORX=49TO57:SET(X,Y):NEXTX:Y=38:FORX=22TO30:SET(X,Y):NEXTX:Y=38:FORX=49TO57:SET(X,Y):NEXTX:
290 Y=40:FORX=26TO69:SET(X,Y):NEXTX:Y=6:FORX=22TO30:SET(X,Y):NEXTX:Y=6:FORX=49TO57:SET(X,Y):NEXTX:X=39:FORY=3TO4:SET(X,Y):NEXTY:X=22
:FORY=6TO14:SET(X,Y):NEXTY:X=26:FORY=4TO6:SET(X,Y):NEXTY:X=26:FORY=14TO30:SET(X,Y):NEXTY:
300 X=26:FORY=38TO40:SET(X,Y):NEXTY:X=49:FORY=6TO14:SET(X,Y):NEXTY:X=49:FORY=30TO38:SET(X,Y):NEXTY:X=53:FORY=4TO6:SET(X,Y):NEXTY:X=5
3:FORY=38TO40:SET(X,Y):NEXTY:X=57:FORY=6TO14:SET(X,Y):NEXTY:X=57:FORY=30TO38:SET(X,Y):NEXTY:
310 X=69:FORY=28TO33:SET(X,Y):NEXTY:X=69:FORY=35TO40:SET(X,Y):NEXTY:X=40:FORY=40TO42:SET(X,Y):NEXTY:X=30:FORY=6TO14:SET(X,Y):NEXTY:X
=22:FORY=30TO38:SET(X,Y):NEXTY:X=30:FORY=30TO38:SET(X,Y):NEXTY:
320 FORN=1TO10000:NEXTN:CLS
330 GOTO20
```

Voltage-Regulator Design

5.

THIS final program can be used in the design of three different types of zener-diode voltage-regulator circuits. It was designed and tested with TRS-80 Level II, but is directly compatible with TRS-80 Level I.

The program is applicable to all three of the circuits shown in the diagram. Lines 30 through 70 ask for information common to all three. Lines 100 through 110 are used only in the first circuit. (We really don't care about the diode's impedance in the other circuits since it isn't seen by the load.)

Line 120 asks which circuit is of interest and lines 130 to 150 branch to the respective graphic "subroutines," which draw the circuits. Lines 860, 1060, and 1250 are simple timing clocks to hold the schematic of the regulator on the screen for a specified amount of time. With the value shown, the display remains for about 20 seconds.

After the clock times out, the computer goes to the next line which directs the computer to the proper line number for the start of that particular part of the program. The first few lines of each individual circuit program query the designer for any additional data that is needed. Following these query lines are a number of data-operation or formula lines.

Following these calculations, the computer prints the results. Unfortunately these results are "ideal" values, and it is up to the designer to pick the nearest standard value. Another set of values is calculated with a 50% safety factor.

One change you may want to make in the program is in lines 60 and 70. Instead of entering power in these two lines, you may want to enter the maximum and minimum current. By doing this, you can eliminate lines 80 and 90. (However, use "J" as the variable in line 60 and "K" in line 70 so that you don't have to rewrite these factors in the remainder of the program. ◇



The program here is applicable to any one of these circuits.

```
10 CLS
20 PRINT@16,"VOLTAGE REGULATOR CIRCUITS"
30 INPUT"ENTER REGULATED OUTPUT VOLTAGE DESIRED";A
40 INPUT"ENTER MINIMUM VOLTAGE FROM POWER SUPPLY FILTER";B
50 INPUT"ENTER MAXIMUM VOLTAGE FROM POWER SUPPLY FILTER";C
60 INPUT"ENTER MAXIMUM POWER OF LOAD";D
70 INPUT"ENTER MINIMUM POWER OF LOAD";E
80 J=D/A
90 K=E/A
100 L=A/.1*J
110 F=A/.1*K
120 INPUT"ENTER CIRCUIT DESIRED 1 , 2 OR 3";M
130 IF M=1 THEN 720
140 IF M=2 THEN 880
150 IF M=3 THEN 1080
160 CLS
170 PRINT"THE CIRCUIT YOU HAVE PICKED WILL REQUIRE ONE ZENER DIODE AND ONE SERIES RESISTOR. REFER TO CIRCUIT #1."
180 G=(C-A)/(1.1*J)
190 I=(C-A)*(1.1*J)
200 H=A*(J-K)
210 PRINT"CALCULATED VALUE OF R1 IS";G;"OHMS (PICK NEXT STANDARD LOWER VALUE)"
220 PRINT"CALC. WATTAGE OF R1 IS";I;"WATTS"
230 PRINT"CALC. VALUE OF D1 IS";A;"VOLTS @";H;"WATTS"
240 PRINT"MAX. LOAD CURRENT IS";J;"AMPS."
250 PRINT"MIN. LOAD CURRENT IS";K;"AMPS."
260 PRINT"MIN. DIODE IMPEDANCE IS";F;"OHMS"
270 PRINT"MAX. DIODE IMPEDANCE IS";F;"OHMS"
275 PRINT"      <>        <>        <>        <>    "
280 PRINT"FOR A SAFETY FACTOR OF 50%:"
290 PRINT"R1 SHOULD EQUAL ";G;"OHMS AT";1.5*I;"WATTS"
300 PRINT"ZENER DIODE D1 SHOULD EQUAL";A;"VOLTS AT";1.5*H;"WATTS"
310 END
320 PRINT"THE CIRCUIT YOU HAVE CHOSEN WILL REQUIRE 1 ZENER DIODE, 1 SERIES RESISTOR , AND 1 TRANSISTOR. REFER TO CIRCUIT #2."
330 INPUT"ENTER BETA OF Q1";N
340 INPUT"ENTER IF Q1 IS SILICON=1, GERMANIUM=2";L
350 I=J/N
360 IF L=1 THEN H=A+.6 ELSE H=A+.3
370 G=(B-H)/I
380 O=(B-H)*I[2
390 P=(C-H)/G
400 Q=H*P
410 R=C-A
420 S=J*R
421 PRINT"RESISTOR R1 CALCULATES TO";G;"OHMS (PICK NEXT LOWEST STANDARD VALUE),AT";O;"WATTS"
422 PRINT"ZENER DIODE D1 CALCULATES TO";H;"VOLTS AT";Q;"WATTS"
423 PRINT"TRANSISTOR Q1 CALCULATES TO";R;"VOLTS (VCE),AT";S;"WATTS"
430 PRINT"   <>          <>          <>          "
440 PRINT"FOR A SAFETY FACTOR OF 50%:"
450 PRINT"R1 SHOULD BE";1.5*O;"WATTS @ ";G;"OHMS"
460 PRINT"ZENER D1 SHOULD BE";1.5*Q;"WATTS @ ";H;"VOLTS"
470 PRINT"Q1 SHOULD BE";1.5*S;"WATTS @ ";1.5*R;"VOLTS"
480 END
490 CLS
500 PRINT"THE CIRCUIT YOU HAVE CHOSEN WILL REQUIRE 1 ZENER DIODE , 1 RESISTOR , AND TWO TRANSISTORS. REFER TO CIRCUIT # 3"
510 INPUT"ENTER THE BETA OF Q1";N
520 INPUT"ENTER THE BETA OF Q2";T
530 INPUT"ENTER IF Q1 & Q2 ARE SILICON=1 , GERMANIUM=2";L
540 IF L=1 H=A+1.2 ELSE H=A+.6
550 U=N*T
560 I=J/N
570 V=C-A
580 S=I/T
590 G=(C-H)/S
600 W=(C-H)/G
610 Q=H*W
620 O=(C-H)*I[2
630 P=C-A
640 R=(C-A)*J
650 X=(C-A)*I
661 PRINT"THE CALC. VALUE OF R1 IS";G;"OHMS (PICK THE NEXT STANDARD LOWER VALUE)"
662 PRINT"THE CALC. POWER OF R1 IS";O;"WATTS"
663 PRINT"THE CALC. VALUE OF D1 IS";H;"VOLTS @";Q;"WATTS"
664 PRINT"THE CALC. VALUE OF Q1 IS";R;"WATTS @ A VCE OF";P;"VOLTS"
665 PRINT"THE CALC. VALUE OF Q2 IS";X;"WATTS @ A VCE OF";(C-A);"VOLTS"
667 PRINT"   <>          <>          <>          <>    "
668 PRINT"FOR A SAFETY FACTOR OF 50%:"
670 PRINT"R1 SHOULD BE";1.5*O;"WATTS @ ";G;"OHMS"
680 PRINT"Z1 SHOULD BE";1.5*Q;"WATTS @ ";H;"VOLTS"
690 PRINT"Q1 SHOULD BE";1.5*R;"WATTS @ A VCE OF";1.5*P;"VOLTS"
700 PRINT"Q2 SHOULD BE";1.5*X;"WATTS @ A VCE OF";1.5*(C-A);"VOLTS"
710 END
720 CLS
730 PRINT@135,"RESISTOR"
740 PRINT@12,"COPY THIS CIRCUIT DOWN"
750 PRINT@67,"R1"
760 PRINT@454,"ZENER DIODE D1"
770 PRINT@490,"LOAD"
780 PRINT@320,"UNREGULATED INPUT VOLTAGE"
790 PRINT@499,"REGULATED"
800 PRINT@564,"OUTPUT"
810 Y=7:FORX=6TO12:SET(X,Y):NEXTX:Y=5:FORX=12TO32:SET(X,Y):NEXTX:Y=9:FORX=12TO32:SET(X,Y):NEXTX:Y=7:FORX=32TO88:SET(X,Y):NEXTX:Y=37:
FORX=7TO88:SET(X,Y):NEXTX:Y=32:FORX=80TO98:SET(X,Y):NEXTX:Y=15:FORX=80TO98:SET(X,Y):NEXTX:
820 Y=26:FORX=44TO64:SET(X,Y):NEXTX:Y=21:FORX=44TO64:SET(X,Y):NEXTX:
830 X=12:FORY=5TO9:SET(X,Y):NEXTY:X=32:FORY=5TO9:SET(X,Y):NEXTY:X=88:FORY=7TO15:SET(X,Y):NEXTY:X=80:FORY=15TO32:SET(X,Y):NEXTY:X=54:
FORY=7TO20:SET(X,Y):NEXTY:X=54:FORY=26TO37:SET(X,Y):NEXTY:X=98:FORY=15TO32:SET(X,Y):NEXTY:
840 X=43:FORY=21TO26:SET(X,Y):NEXTY:X=88:FORY=33TO37:SET(X,Y):NEXTY:X=64:FORY=21TO26:SET(X,Y):NEXTY:
850 PRINT@826,"CRKT#1"
860 FORN=1TO10000:NEXTN:
870 GOTO160
880 CLS
885 PRINT @ 12, "COPY THIS CIRCUIT DOWN"
890 PRINT@80,"TRANSISTOR Q1"
900 PRINT@390,"RESISTOR R1"
910 PRINT@462,"ZENER D1"
920 PRINT@200,"COLLECTOR"
930 PRINT@285,"BASE"
940 PRINT@226,"EMITTER"
950 PRINT@512,"UNREGULATED INPUT"
960 PRINT@542,"REGULATED OUTPUT VOLTAGE"
970 Y=8:FORX=6TO35:SET(X,Y):NEXTX:Y=6:FORX=35TO65:SET(X,Y):NEXTX:Y=10:FORX=35TO65:SET(X,Y):NEXTX:Y=8:FORX=65TO88:SET(X,Y):NEXTX:Y=15
:FORX=10TO16:SET(X,Y):NEXTX:
980 Y=32:FORX=6TO88:SET(X,Y):NEXTX:X=35:FORY=6TO10:SET(X,Y):NEXTY:X=65:FORY=6TO10:SET(X,Y):NEXTY:X=10:FORY=8TO15:SET(X,Y):NEXTY:X=16
:FORY=13TO17:SET(X,Y):NEXTY:X=32:FORY=13TO17:SET(X,Y):NEXTY:
990 X=50:FORY=10TO18:SET(X,Y):NEXTY:X=46:FORY=18TO26:SET(X,Y):NEXTY:X=55:FORY=18TO26:SET(X,Y):NEXTY:X=50:FORY=26TO32:SET(X,Y):NEXTY:
1000 Y=13:FORX=16TO31:SET(X,Y):NEXTX:
1010 Y=17:FORX=16TO31:SET(X,Y):NEXTX:
1020 Y=18:FORX=46TO55:SET(X,Y):NEXTX:
1030 Y=26:FORX=47TO55:SET(X,Y):NEXTX:
1040 Y=15:FORX=32TO50:SET(X,Y):NEXTX:
1050 PRINT@824,"CRKT#2"
1060 FORN=1TO10000:NEXTN:
1065 CLS
1070 GOTO320
1080 CLS
1085 PRINT @ 5, "COPY THIS CIRCUIT DOWN"
1090 PRINT@396,"R1"
1100 PRINT@620,"REGULATED OUTPUT"
1110 PRINT@36,"TRANSISTOR Q1"
1120 PRINT@640,"UNREGULATED INPUT"
1130 PRINT@180,"EMITTER"
1140 PRINT@222,"Q2"
1150 PRINT@237,"BASE"
1160 PRINT@279,"C"
1170 PRINT@416,"BASE"
1180 PRINT@590,"ZENER DIODE D1"
1190 Y=5:FORX=6TO72:SET(X,Y):NEXTX:Y=3:FORX=73TO100:SET(X,Y):NEXTX:Y=5:FORX=101TO111:SET(X,Y):NEXTX:Y=7:FORX=73TO100:SET(X,Y):NEXTX:
Y=20:FORX=29TO47:SET(X,Y):NEXTX:Y=24:FORX=29TO47:SET(X,Y):NEXTX:Y=17:FORX=50TO75:SET(X,Y):NEXTX:Y=22:FORX=20TO27:SET(X,Y):NEXTX:
1200 Y=13:FORX=50TO75:SET(X,Y):NEXTX:Y=15:FORX=76TO87:SET(X,Y):NEXTX:Y=15:FORX=40TO49:SET(X,Y):NEXTX:Y=26:FORX=58TO67:SET(X,Y):NEXTX
:Y=34:FORX=58TO67:SET(X,Y):NEXTX:Y=40:FORX=6TO111:SET(X,Y):NEXTX:Y=22:FORX=20TO28:SET(X,Y):NEXTX:
1210 X=20:FORY=5TO22:SET(X,Y):NEXTY:X=29:FORY=20TO24:SET(X,Y):NEXTY:X=40:FORY=6TO15:SET(X,Y):NEXTY:X=47:FORY=20TO24:SET(X,Y):NEXTY:X
=50:FORY=13TO17:SET(X,Y):NEXTY:X=58:FORY=26TO34:SET(X,Y):NEXTY:X=67:FORY=26TO34:SET(X,Y):NEXTY:
1220 X=62:FORY=18TO25:SET(X,Y):NEXTY:X=75:FORY=13TO17:SET(X,Y):NEXTY:X=87:FORY=8TO15:SET(X,Y):NEXTY:X=73:FORY=3TO7:SET(X,Y):NEXTY:X=
100:FORY=3TO7:SET(X,Y):NEXTY:
1230 Y=22:FORX=48TO62:SET(X,Y):NEXTX:X=62:FORY=35TO40:SET(X,Y):NEXTY:
1240 PRINT@930,"CRKT#3"
1250 FORN=1TO10000:NEXTN:
1260 GOTO490
```